15-150 Fall 2019
Lecture 18:
Sequences in practice

Stephen Brookes *

*Building on notes by Dan Licata
1 Sequences

Last time we introduced the signature \texttt{SEQ} and told you about a structure \texttt{Seq} that implements this signature, and provides parallel-friendly sequence types and operations.

signature \texttt{SEQ} =

sig
  type 'a seq
  exception Range
  val length : 'a seq -> int
  val nth : int -> 'a seq -> 'a
  val tabulate : (int -> 'a) -> int -> 'a seq
  val empty : unit -> 'a seq
  val map : ('a -> 'b) -> ('a seq -> 'b seq)
  val reduce : ('a * 'a -> 'a) -> 'a -> 'a seq -> 'a
  val mapreduce : ('a -> 'b) -> 'b -> ('b * 'b -> 'b) -> 'a seq -> 'b
end

For today, assume that we’ve open-ed this structure, so we don’t have to use qualified names to get at the components. When developing code it might be preferable to use fully qualified names — this may help you keep track of where the components come from. But in class we’re going to work within a clearly defined framework, and we’ll use the more succinct notation.

In today’s lecture we’ll discuss some examples using sequences to solve some non-trivial problems. We won’t necessarily have time in class to prove correctness and/or analyze the work and span of each function. Some of these may be tasks in lab, and some may appear on homework or exams. We’d like you to get in the habit of sketching correctness proofs and work/span analysis as a matter of course! We did present work/span specs in the previous lecture for the main sequence operations, and we used them there to analyze a code fragment for summing integers in a sequence of integer sequences. You can use similar techniques here.
2 Gravitational simulation

We can use sequences to perform an *n-body simulation* of the effect of gravity on celestial bodies. Given the mass, position, and velocity of *n* celestial bodies, we want to simulate their movement over time due to gravitational forces. Such *n*-body simulations are used by astrophysicists — for example, their results provide evidence that there is a black hole in the center of the Milky Way. Similar ideas apply in other settings involving other forces; for example, Coulomb’s law has a similar mathematical form to Newton’s law of gravitation, and is used in simulations of protein folding.

Here, we consider the *n*-body problem in a 2-dimensional universe. The entire development can easily be generalized to 3-dimensions, but it is much easier to draw pictures on a 2-dimensional plane. The key ingredient in our *n*-body simulation is involves computing the acceleration on one body due to the combined gravitational attraction of a collection of other bodies. For a universe consisting of *n* bodies this requires computing a quadratic number \(O(n^2)\) of pairwise interactions. However, this problem is highly parallelizable, as the acceleration on one body due to one other body can be computed independently of the remaining bodies; and each body can compute its own combined gravitational acceleration in parallel.

(The following explanatory paragraphs are based on Wikipedia.)

Newton’s law of universal gravitation states that every point mass in the universe attracts every other point mass with a force that is directly proportional to the product of their masses and inversely proportional to the square of the distance between them. (Separately it was shown that large spherically symmetrical masses attract and are attracted as if all their mass were concentrated at their centers.) This is a general physical law derived from empirical observations by what Newton called “induction”. (Not the same as the proof method that we call induction!) Newton’s law was formulated in the famous work Philosophiae Naturalis Principia Mathematica (colloquially known as “the Principia”), first published in 1687.

In modern language, the law states the following:

A point mass attracts another point mass by a force directed along the line between the points. The magnitude of the force is proportional to the product of the two masses and inversely proportional to the square of the distance between them.
As a formula, 
\[ F = \frac{G m_1 m_2}{r^2} \]
where \( F \) is the magnitude of the force between the masses, \( G \) is a gravitational constant, \( m_1 \) is the first mass, \( m_2 \) is the second mass, and \( r \) is the distance between the centers of the masses.

Newton’s Law of Motion says that the acceleration on a body with mass \( m \) due to a force of magnitude \( F \) (in some direction) is (a vector in the same direction, with magnitude) \( F/m \). This is usually abbreviated as “force = mass times acceleration”, or \( F = ma \). So the acceleration on a body of mass \( m \) due to the gravitational attraction of another body of mass \( m_2 \) is
\[ a = \frac{G m_2}{r^2} \]
where \( r \) is the distance between the bodies. \(^1\)

We’re thinking of “vectors” here as pairs \((x, y)\) of real numbers, and such a pair determines a “direction” (all lines with the slope \( y/x \)) and a magnitude (the square root of \( x^2 + y^2 \)). For example, the vector \((1.0, 1.0)\) has slope 1.0, so is directed at 45% (north-west), and has magnitude \( \sqrt{2} \).

We’re also thinking of points as \((x, y)\) pairs — the “\(x\)”-coordinate and the “\(y\)”-coordinate; the “origin” is the point \((0.0, 0.0)\).

Given a point \((x, y)\) and a vector \((dx, dy)\), we can displace the point along the vector and obtain a new point: this will be the point \((x + dx, y + dy)\).

Two vectors can be added together to produce a vector. In math we usually use \(+\) to mean either vector addition or real number addition, relying on the context to disambiguate. So in math we might write
\[ (x_1, y_1) + (x_2, y_2) = (x_1 + x_2, y_1 + y_2) \]
for the sum of two vectors \((x_1, y_1)\) and \((x_2, y_2)\).

Given two points \((a_1, b_1)\) and \((a_2, b_2)\) the vector that represents the line from \((a_2, b_2)\) to \((a_1, b_1)\) is obtained by subtraction, and again using math notation we might write
\[ (a_2, b_2) - (a_1, b_1) = (a_2 - a_1, b_2 - b_1). \]

\(^{1}\)The gravitational acceleration on a body is independent of the body’s mass, an observation usually attributed to Galileo. There is a (probably apocryphal) story about Galileo dropping balls of different masses off the top of the Leaning Tower of Pisa to demonstrate this, and contradicting conventional wisdom that dated back to Aristotle, who thought that more massive objects fall faster.
With these ingredients in place, we now have enough to explain how to implement some simple ML functions for use in developing an \(n\)-body simulation.

The gravitational constant \(G\) is approximately \(6.67 \times 10^{-11} \text{N}(m/\text{kg})^2\), i.e. when the force is measured in units known as Newtons, the masses are in kilograms, and the distance is in meters, the value of \(G\) is about 0.0000000000667. To keep the math simple and avoid dealing with very small real numbers we’ll assume that \(G\) is 1.0. This amounts (basically) to scaling all of our data appropriately.

In developing our simulation we will ignore what happens when bodies collide; actually when bodies get very close in space other forces than gravity may come into effect. So we’ll design our code so that when objects get “too close” we ignore gravity.

**Points and vectors**

As we said, we will represent points in the plane by pairs of real numbers. And we represent vectors as pairs of real numbers. The velocity and the acceleration of a body will be modeled as vectors, and will be used to compute the displacement to apply when “stepping” the universe for a single time step.

type point = real * real
type vect = real * real

Even though these two types are the same, it makes sense to use different names; values of type point and values of type vect play different roles in the above narrative, and we want our ML implementation to be faithful to that description.

We need the following operations on points and vectors.

- Vectors have a zero, addition, and scalar multiplication:

  ```ml
  val zero : vect = (0.0, 0.0)
  fun add ((x1,y1):vect, (x2,y2):vect):vect = (x1 + x2, y1 + y2)
  fun scale(c:real, (x,y):vect):vect = (c * x, c * y)
  ```

- To compute the magnitude of a vector:

  ```ml
  fun mag ((x,y) : vect) : real = Math.sqrt (x * x + y * y)
  ```
• Given two points, we can compute the vector from the first to the second: (note the order of subtraction)

\[
\text{fun diff ((x1,y1):point, (x2,y2):point) : vect = (x2 - x1, y2 - y1)}
\]

• Note that the relative distance of two points \( p \) and \( q \), the distance between them, is simply

\[
\text{mag(diff(p, q))}
\]

• To displace a point by a vector:

\[
\text{fun displace ((x,y):point, (x’,y’):vect) : point = (x + x’, y + y’)}
\]

• To check if two points are too close for comfort we’ll use a parameter (tweakable, depending on the data), which we’ll take as 0.1, and say that points have collided when their relative distance is less than this parameter:

\[
\text{fun small (r : real) : bool = (Real.abs r < 0.1)}
\]

• In summary, we have the following operations on vectors:

\[
\begin{align*}
\text{type point} & = \text{real} \times \text{real} \\
\text{type vect} & = \text{real} \times \text{real} \\
\text{val diff: point} & \times \text{point} \to \text{vect} \\
\text{val small : real} & \to \text{bool} \\
\text{val displace : point} & \times \text{vect} \to \text{point} \\
\text{val add : vect} & \times \text{vect} \to \text{vect} \\
\text{val scale : real} & \times \text{vect} \to \text{vect} \\
\text{val zero : vect} & \\
\text{val mag : vect} & \to \text{real}
\end{align*}
\]
Representing celestial bodies

A body is represented by its position (a point), mass (a real number), and velocity (a vector).

(* body = position, mass, velocity *)
type body = point * real * vect

Accelerations

• The acceleration on body \( b \) due to the gravitational attraction of body \( b' \) is given by \( \text{accel} \ b \ b' \), as in:

\[
\text{fun accel (p1, _, _) (p2, m2, _) =
  let
    val d = diff(p1, p2)
    val r = mag d
  in
    if (small r) then zero else scale(G * m2/(r * r * r) , d)
  end;
\]

(See how the ML code corresponds clearly to the intuitive description given earlier. We scale \( d \) using the inverse cube of its magnitude \( r \), because that produces the same result as a vector of magnitude \( 1/r^2 \). Note the use of wildcard patterns to indicate that the acceleration doesn’t depend on the mass of the first body, as Galileo said. Also we have curried this function, to make it easy to use by partially applying it – to a given “first” body.)

• The acceleration on body \( b \) due to the collective action of the bodies in a sequence \( s \) is \( \text{accels} \ b \ s \). We can calculate this vector by mapping \( \text{accel} \ b \) along \( s \) and adding the gravitational contributions of each body in the sequence. Rather than a map followed by a reduce, it’s very simple to do a mapreduce:

\[
\text{fun accels b s = mapreduce (accel b) zero add s;}
\]
Moving

Suppose we have a body \( b = (p, m, v) \), and we have computed the acceleration on \( b \) to be the vector \( a \). For a small timestep \( dt \) we can approximate the motion of this point during this time step by assuming that the acceleration stays constant. Then, according to standard Newtonian mechanics, at the end of this time interval the body will have been displaced by the (vector) sum of \( v dt \) and \( \frac{1}{2}a dt^2 \). (Here of course \( v dt \) is the vector obtained by scaling \( v \) by \( dt \), and \( \frac{1}{2}a dt^2 \) is the vector obtained by scaling \( a \) by \( \frac{1}{2}dt^2 \).) After this time step the body will have the same mass, a new position obtained by this displacement, and a new velocity obtained by adding \( a dt \) to \( v \). Hence,

\[
(*) \text{ move : body } \rightarrow \text{ vect } \times \text{ real } \rightarrow \text{ body } *
\]

fun move (p, m, v) (a, dt) =
let
  val p' = displace(p, add(scale(dt, v), scale(0.5 * dt * dt, a)))
  val v' = add(v, scale(dt, a))
in
  (p', m, v')
end;

And we can again use a higher-order function to do parallel movement of a collection of bodies:

\[
(*) \text{ step : body seq } \rightarrow \text{ real } \rightarrow \text{ body seq } *
\]

fun step s dt = map (fn b => move b (accels b s, dt)) s;
An example

val sun = ((0.0,0.0), 332000.0, (0.0,0.0));
val earth = ((1.0, 0.0), 1.0, (0.0,18.0));

Assume for simplicity that the sun is so much more massive than the Earth
that we expect it to stay pretty much fixed in space while the Earth revolves
in orbit. And we’re interested in the trajectory of the Earth, i.e. a list of
points that represent successive snapshots of the Earth’s progression around
its orbit, taken in successive time steps.

val s : body seq = tabulate (fn 0 => sun | 1 => earth | _ => raise Range) 2;
val s = [((0.0,0.0),332000.0,(0.0,0.0)),((1.0,0.0),1.0,(0.0,18.0))]: body seq

step s 0.01;
val it = [((5E~05,0.0),332000.0,(0.01,0.0)),((-15.6,0.18),1.0,(-3320.0,18.0))]: body seq

(Note that the sun has hardly moved, but the Earth has.)

We can generate orbits using:

(* orbit : body -> int * real -> point list *)
(* orbit b (n, dt) = n steps of motion of b under gravity of sun *)

fun orbit b (n, dt) =
  if n=0 then [ ] else
    let
      val (p’, m, v’) = move b (accel b sun, dt)
    in
      p’ :: orbit (p’, m, v’) (n-1, dt)
    end;

orbit earth (10, 0.01);
val it =
[(-15.6,0.18),(-48.7318019171,0.359213099043),
 (-81.7884162248,0.537587775754),(-114.835559608,0.71589462109),
 (-147.878962872,0.894177309445),(-180.920348361,1.07244756107),
 (-213.960467679,1.25071021688),(-246.999717285,1.42896774708),
 (-280.03833222,1.60722158368),(-313.076463412,1.78547263142)]: point list
The work of `accels` is $O(n^2)$. The span is $O(\log n)$: the only chain of dependencies is adding up the components of the acceleration.

Thus, this algorithm is highly parallelizable. Where did this parallelism come from? First, from expressing the algorithm with maps and reduces on sequences, rather than lists. Second, from expressing the inner computations as mathematical calculations (in this case, calculations on real numbers), which are all independent of one another. We hope you can see the close correspondence between the math in question and the code, which is important both for parallelism, and for elegance/readability of the code.
3 Self-test

1. Write an ML function

    \texttt{center : (point * mass) seq -> point * mass}

such that \texttt{center s} is equal to the center of mass of \texttt{s}. If \texttt{s} is the sequence \langle (p_1, m_1), \ldots, (p_n, m_n) \rangle the center of mass is \((p, M)\) where

\[
\begin{align*}
M &= m_1 + \cdots + m_n \quad \text{(real addition)} \\
p &= (1/M)(m_1p_1 + \cdots + m_np_n) \quad \text{(vector addition and scaling)}
\end{align*}
\]

We write \(m_ip_i\) for the point obtained by multiplying the coordinates of \(p_i\) by \(m_i\). Use \texttt{add, scale, map, reduce,} and/or \texttt{mapreduce}.

2. Explain why \texttt{add} is an associative function, with \texttt{zero} as its identity element.

3. Let \texttt{s} be a value of type \texttt{vect seq}. Discuss the types, behavior, and work/span properties of the following expressions:

   (i) \texttt{map (fn v => scale(2.0, v))}
   
   (ii) \texttt{map (fn v => scale(2.0, v)) s}

4. What changes would you make to the code developed above if you wanted to model a 3-dimensional universe? What effects would this have on the work and span of the various operations, specifically for the 3D-versions of \texttt{accels} and \texttt{step}? How about the analogous operations on lists?